

## Soluzione di un sistema 3x4 con Gauss

Enrico Bertolazzi  
Esercitazioni Maple di Geometria

```
> # carica la libreria per l'algebra lineare  
with(LinearAlgebra) ;  
> # definisce il termine noto  
b := <1,1,2> ;
```

$$b := \begin{bmatrix} 1 \\ 1 \\ 2 \end{bmatrix} \quad (1)$$

```
> # definisce la matrice dei coefficienti + termine noto  
A := <<3,3,3>|<1,2,3>|<-1,0,0>|<1,0,8>|b>;
```

$$A := \begin{bmatrix} 3 & 1 & -1 & 1 & 1 \\ 3 & 2 & 0 & 0 & 1 \\ 3 & 3 & 0 & 8 & 2 \end{bmatrix} \quad (2)$$

```
> # mette a 1 A[1,1]  
D1 := DiagonalMatrix(<1/3,1,1>);  
D1A := D1.A ;
```

$$D1 := \begin{bmatrix} \frac{1}{3} & 0 & 0 \\ 0 & 1 & 0 \\ 0 & 0 & 1 \end{bmatrix} \quad (3)$$

$$D1A := \begin{bmatrix} 1 & \frac{1}{3} & -\frac{1}{3} & \frac{1}{3} & \frac{1}{3} \\ 3 & 2 & 0 & 0 & 1 \\ 3 & 3 & 0 & 8 & 2 \end{bmatrix}$$

```
> # prima matrice per azzerare gli elementi [2,1] e [3,1]  
L1 := <<1,-3,-3>|<0,1,0>|<0,0,1>> ;
```

$$L1 := \begin{bmatrix} 1 & 0 & 0 \\ -3 & 1 & 0 \\ -3 & 0 & 1 \end{bmatrix} \quad (4)$$

```
> # applica la trasformazione  
L1D1A := L1.D1A ;
```

$$L1D1A := \begin{bmatrix} 1 & \frac{1}{3} & -\frac{1}{3} & \frac{1}{3} & \frac{1}{3} \\ 0 & 1 & 1 & -1 & 0 \\ 0 & 2 & 1 & 7 & 1 \end{bmatrix} \quad (5)$$

> # seconda matrice per azzerare gli elementi [1,2] [3,2]  
 L2 := <<1,0,0>|<-1/3,1,-2>|<0,0,1>> ;

$$L2 := \begin{bmatrix} 1 & -\frac{1}{3} & 0 \\ 0 & 1 & 0 \\ 0 & -2 & 1 \end{bmatrix} \quad (6)$$

> # applica la trasformazione  
 L2L1D1A := L2.L1D1A ;

$$L2L1D1A := \begin{bmatrix} 1 & 0 & -\frac{2}{3} & \frac{2}{3} & \frac{1}{3} \\ 0 & 1 & 1 & -1 & 0 \\ 0 & 0 & -1 & 9 & 1 \end{bmatrix} \quad (7)$$

> # mette a 1 [3,3]  
 D2 := DiagonalMatrix(<1,1,-1>);  
 D2L2L1D1A := D2.L2L1D1A ;

$$D2 := \begin{bmatrix} 1 & 0 & 0 \\ 0 & 1 & 0 \\ 0 & 0 & -1 \end{bmatrix} \quad (8)$$

$$D2L2L1D1A := \begin{bmatrix} 1 & 0 & -\frac{2}{3} & \frac{2}{3} & \frac{1}{3} \\ 0 & 1 & 1 & -1 & 0 \\ 0 & 0 & 1 & -9 & -1 \end{bmatrix}$$

> # terza matrice per azzerare gli elementi [1,3] [2,3]  
 L3 := <<1,0,0>|<0,1,0>|<2/3,-1,1>>;

$$L3 := \begin{bmatrix} 1 & 0 & \frac{2}{3} \\ 0 & 1 & -1 \\ 0 & 0 & 1 \end{bmatrix} \quad (9)$$

> # applica la trasformazione  
 L3D2L2L1D1A := L3.D2L2L1D1A ;

$$L3D2L2L1D1A := \begin{bmatrix} 1 & 0 & 0 & -\frac{16}{3} & -\frac{1}{3} \\ 0 & 1 & 0 & 8 & 1 \\ 0 & 0 & 1 & -9 & -1 \end{bmatrix} \quad (10)$$

> # il sistema ora e` immediatamente risolubile  
 RES := L3D2L2L1D1A.<x1,x2,x3,x4,-1> ;

$$RES := \begin{bmatrix} x1 + \frac{1}{3} - \frac{16}{3} x4 \\ -1 + x2 + 8 x4 \\ 1 + x3 - 9 x4 \end{bmatrix} \quad (11)$$

> isolate(RES[1],x1) ;  
 isolate(RES[2],x2) ;  
 isolate(RES[3],x3) ;

$$\begin{aligned} x1 &= -\frac{1}{3} + \frac{16}{3} x4 \\ x2 &= 1 - 8 x4 \\ x3 &= -1 + 9 x4 \end{aligned} \quad (12)$$